اسکیم (زبان برنامه‌نویسی)

بندانگشتی|نمونه‌ای از برنامه نویسی با اسکیم
**اسکیم**  یک زبان برنامه‌نویسی تابعی است که از چندین پارادایم مثل پارادایم تابعی و دستوری پشتیبانی می کند. این زبان یکی از سه ویش های لیسپ در کنار Common Lisp و Clojure می باشد. برعکس Common Lisp، اسکیم از فلسفه طراحی حداقل گرا استفاده می کند، بر اساس این فلسفه هسته استاندارد باید کوچک باشد، ضمن این که برای توسعه زبان ابزار های قدرتمندی را دارا باشد.

اسکیم در طول دهه ۷۰ در آزمایشگاه هوش مصنوعی دانشگاه MIT توسط توسعه دهندگانش، Guy L. Steele و Gerald Jay Sussman به وسیلهٔ یک سری از یادداشت‌ها که حالا به عنوان مقالات لاندا شناخته می‌شوند توسعه و انتشار داده شد. این زبان از اولین زبان‌های برنامه‌نویسی بود که از تداوم کلاس اول پشیبانی می‌کند. تأثیر قابل توجهی بر تلاشی که منجر به توسعه Lisp رایج شد داشت.

زبان اسکیم در استاندارد IEEE رسمی استانداردسازی شده‌است و عملاً استانداردی که گزارش تجدید نظر در طرح زبان الگوریتمی اسکیم (RnRs) نامیده می‌شود. بزرگترین استاندارد پیاده‌سازی شده R5RS است (۱۹۹۸);  یک استاندراد جدید، R6RS، در سال ۲۰۰۷ تصویب شده‌است.اسکیم پایه کاربری متنوعی را به دلیل فشردگی و ظرافت دارد، اما فلسفه حداقل گرای آن باعث واگرایی میان پیاده‌سازی‌های کاربردی آن شده‌است، به‌طوری‌که کمیته فرمان اسکیم آن را «سیار ترین زبان برنامه‌نویسی» و «یک خانواده از گویش ها» می‌نامد تا یک زبان برنامه‌نویسی تنها.

## تاریخچه
مقاله اصلی: History of the Scheme programming language

### ریشه‌ها
اسکیم در دهه ۷۰ به عنوان تلاشی برای فهمیدن مدل بازیگر Carl Hewitt، برای قصدی که Steele و Sussmanدر "مفسر کوچک Lisp" با استفاده از Maclisp نوشتند و بعد از آن "مکانیزم‌های افزوده شده برای ساختن بازیگرها و ارسال پیام هاً شروع شد. اسکیم در اصل و به رسم دیگر زبان‌های مشتق شده از Lisp مانند Planner یا Connover , "schemer" نامیده می‌شد. نام حال حاضر نتیجه از استفاده نویسندگان از سیستم عامل ITS است، که نام فایل‌ها را به دو بخش از که هر کدام حد اکثر شش کارکتر می‌پذیرند سات. در حال حاضر، "Schemer" برای اشاره به برنامه‌نویس اسکیم استفاده می‌شود.

## ویژگی‌های متمایز
اسکیم در درجه اول یک زبان برنامه‌نویسی تابعی است. این زبان، ویژگی‌های بسیاری را با دیگر زبان‌های خانواده زبان لیسپ (Lisp) دارد. نحو بسیار ساده اسکیم بر پایهٔ عبارات، لیست‌های پرانتزی است که در آن یک عملگر پیشوند با آرگومانهایش دنبال می‌شود. بدین ترتیب، برنامه‌های اسکیم شامل دنبالهٔ لیست‌های تو در تو است. لیست‌ها همچنین، ساختار اصلی داده‌ها در اسکیم هستند، که منجر به یک هم‌ارزی نزدیک بین کد منبع و فرمت‌های داده (homoiconicity) می‌شود. برنامه‌های اسکیم می‌توانند به راحتی بخش‌های کد اسکیم را به صورت پویا ایجاد و ارزیابی کنند.

وابستگی به لیست‌ها به عنوان ساختار داده‌ها بین همه لهجه‌های لیسپ وجود دارد. اسکیم، یک مجموعه غنی از اولویت‌های پردازش لیست از جمله "cons", "car" و "cdr" را از نمونه‌های قدیمی تر لیسپ (اجداد لیسپ) به ارث می‌برد. اسکیم به شدت اما به صورت پویا از متغیرهای تایپ شده‌استفاده می‌کند و از رویه‌های (procedures) کلاس اولیه پشتیبانی می‌کند؛ بنابراین، رویه‌ها را می‌توان به عنوان مقادیر به متغیرها یا آرگومان‌ها تعیین نمود.

این بخش به‌طور عمده بر ویژگی‌های نوآورانه این زبان تمرکز می‌کند، از جمله ویژگی‌هایی که اسکیم را از سایر لیسپ‌ها متفاوت می‌کند. همچنین به جز موارد ذکر شده، توصیف ویژگی‌های مربوط به استاندارد R5RS.

*در نمونه‌هایی که در این بخش ارائه شده است، از علامت "(نتیجه) result ===>"، برای نشان دادن نتیجهٔ ارزیابی عبارت خط ماقبل استفاده می‌شود. این همان قرارداد استفاده شده در R5RS است.*

## ویژگی‌های طراحی بنیادی

این بخش، آن ویژگی‌های اسکیم را توصیف می‌کند که آن را از دیگر زبان‌های برنامه‌نویسی (از همان ابتدا) متمایز می‌کند. این‌ها جنبه‌هایی از اسکیم هستند که بیشترین تأثیر را بر هر محصول زبان اسکیم دارند، و جنبه‌هایی هستند که تمامی نسخه‌های نسخه‌های زبان برنامه‌نویسی اسکیم، از سال ۱۹۷۳ به بعد، آن را دارا هستند.

## مینیمالیسم (minimalism)

اسکیم یک زبان بسیار ساده است، بسیار آسان‌تر از خیلی از زبان‌های قابل مقایسه دیگر (از نظر قدرت بیان) پیاده‌سازی می‌شود. این سهولت را می‌توان به استفاده از محاسبات لامبدا تا بسیاری از نحو این زبان، از فرم‌های ابتدایی تر نسبت داد. به عنوان مثال از ۲۳ ساختار نحوی مبتنی بر عبارات s در استاندارد R5RS، ۱۱ تا به عنوان فرم مشتق یا فرم کتابخانه‌ای طبقه‌بندی شده‌اند، که می‌تواند به صورت ماکروی شامل اشکال اساسی بیشتر، به ویژه لامبدا نوشته شود. همان‌طور که R5RS عنوان می‌کند (R5RS بخش ۳٫۱): «اساسی‌ترین ساختارهای انقیاد متغیر عبارات لامبدا است، زیرا تمام دیگر ساختارهای انقیاد متغیر را می‌توان به شکل عبارات لامبدا توضیح داد.»

**فرم‌های بنیادی**: define, lambda, if, quote, unquote, unquote-splicing, quasiquote, define-syntax, let-syntax, letrec-syntax, syntax-rules, set!

**فرم‌های کتابخانه ای**: do, let, let*, letrec, cond, case, and, or, begin, named let, delay

مثال: یک ماکرو برای پیاده‌سازی به عنوان یک عبارت لامبدا برای اجرای انقیاد متغیر.
```scheme
(define-syntax let

 (syntax-rules ()

 ((let ((var expr) ...) body ...)

 ((lambda (var ...) body ...) expr ...))))
```


بنابراین استفاده از "let" مانند بالا، پیاده‌سازی اسکیم عبارت "(let ((a 1)(b 2)) (+ b a))" را به عنوان "((lambda (a b) (+ b a)) 1 2)" بازنویسی می‌کند، که کار پیاده‌سازی را به نمونه رویه‌های کدنویسی کاهش می‌دهد.

در ۱۹۹۸، ساسمَن و استیل اشاره کردند که اسکیمِ مینیمالیسم یک هدف طراحی آگاهانه نخواهد بود، بلکه به نوعی نتیجهٔ ناخواسته فرایند طراحی است. «ما در واقع تلاش می‌کردیم چیزی پیچیده و کشف شده را بسازیم، که به‌طور تصادفی چیزی ساختیم که تمام اهداف ما را دربر گرفته بود اما خیلی ساده‌تر از آنچه ما قبلاً در نظر داشتیم. ما متوجه شدیم که محاسبات لامبدا _ یک فرمالیزم ساده و کوچک _ می‌تواند به عنوان هستهٔ یک زبان برنامه‌نویسی قوی و با معنا به کار رود.»

## محدوده واژگانی (Lexical Scope)

همانند بسیاری از زبان‌های برنامه‌نویسی مدرن و برخلاف لیسپ‌های قدیمی تر مانند ماکلیسپ(Maclisp)، اسکیم به صورت واژگانی محدود می‌شود (= در scope قرار می‌گیرد): تمام انقیاد متغیرهای ممکن در یک واحد برنامه می‌تواند با خواندن متن یک واحد برنامه بدون در نظر گرفتن مفاد آن تحلیل کند. این در مقایسه با محدوده پویایی (dynamic scope) است که مشخصه گویش‌های اولیه لیسپ بود؛ به دلیل هزینه‌های پردازش مرتبط با روش‌های جایگزینی متنی اولیه استفاده شده برای پیاده‌سازی الگوریتم‌های محدوده لغوی (Lexical Scoping) در کامپایلرها و مفسران روز. در آن لیسپ‌ها، برای یک مرجع متغیر آزاد درون یک رویه، کاملاً امکان‌پذیر بود که به انقیادهای کاملاً مجزا خارج از رویه (بسته به فراخوانی) منتسب نماید.

انگیزهٔ ترکیب محدوده لغوی(lexical scoping)، که در اوایل دهه ۱۹۷۰ یک مدل محدوده غیرمعمول بود، به نسخه جدیدی از لیسپ، از مطالعات ساسمَن در مورد زبان الگول (ALGOL) نشئت گرفت. او پیشنهاد کرد که مکانیزم‌های محدوده لغوی مشابه الگول کمک می‌کند تا هدف اولیه خود از اجرای مدل بازیگر هِویت (Hewitt’s Actor) در لیسپ را تحقق ببخشند. بینش کلیدی در چگونگی معرفی محدوده لغوی به یک گویش لیسپ، در مقاله لامبدای ساسمَن و استیل ۱۹۷۵ شهرت یافت؛ «اسکیم: یک مفسر برای محاسبات گسترده لامبدا»، که در آن، آن‌ها مفهوم بستار لغوی را تصویب کردند؛ (در صفحه ۲۱)، که در یک یادداشت هوش مصنوعی از جوئل موسِز کسی که این ایده را به پیتر ج لَندین نسبت داده، شرح داده شده بود.

## محاسبات لامبدا

نشانه ریاضیاتی آلونزو چِرچ (Alonzo Church)، محاسبات لامبدا، الهام بخش استفاده از «لامبدا» به عنوان یک کلید واژه برای معرفی رویه، همچنین تأثیر در توسعه تکنیک‌های برنامه‌نویسی تابعی شامل استفاده از توابع درجه بالاتر در لیسپ می‌باشد. اما لیسپ‌های ابتدایی تر، عبارات مناسبی از محاسبات لامبدا نبودند؛ به دلیل رفتار متغیرهای آزادش.

معرفی محدوده لغوی، مسئله را با ایجاد همبستگی (هم‌ارزی) میان برخی اَشکال لامبدا و بیان عملی آن‌ها در یک زبان برنامه‌نویسی کارا حل کرده‌است. ساسمَن و استیل نشان دادند که زبان جدید می‌تواند برای استنتاج تمام معانی اِخباری و دستوری زبان‌های برنامه‌نویسی دیگر مانند الگول و فرترن و محدوده پویا دیگر لیسپ‌ها، با استفاده از عبارات لامبدا، نه به عنوان رویه ساده، بلکه به عنوان «ساختارهای کنترل و اصلاح‌کننده‌های محیط» استفاده شود.https://web.archive.org/web/20160510140804/http://library.readscheme.org/page1.html آن‌ها شیوهٔ continuation-passing را همراه با اولین توصیف اسکیم در اولین مقالات لامبدا معرفی کردند و در مقالات بعدی، قدرت استفاده عملی محاسبات لامبدا را نشان دادند.

## ساختار بلوک

اسکیم، ساختار بلوکی خود را از زبان‌های با ساختار بلوکی قدیمی تر به ارث برده است، به ویژه الگول. در اسکیم، بلوک‌ها با ۳ ساختار انقیاد پیاده‌سازی می‌شود: “let”، “let*” و “letrec”. برای مثال، ساختار زیر یک بلوک ایجاد می‌کند که در آن یک نماد به نام var به شماره ۱۰ محدود می‌شود:
```scheme
(define var "goose")
;; Any reference to var here will be bound to "goose"
(let ((var 10))
;; statements go here. Any reference to var here will be bound to 10.
)
;; Any reference to var here will be bound to "goose"
```


بلوک‌ها می‌توانند تو در تو باشند تا ساختار بلوکی پیچیدهٔ دلخواه با توجه به نیاز برنامه‌نویس ایجاد نماید. استفاده از ساختار بلوک برای ایجاد انقیادها، خطر برخورد فضای نام را کاهش می‌دهد که در غیر این صورت ممکن است اتفاق بیفتد. (در شرایط عدم استفاده از ساختار بلوک)

یک نوع از “let” و “let*”، اجازه انقیادها را می‌دهد تا به متغیرهای تعریف شده قبلی در همان ساختار ارجاع دهد؛ بنابراین:
```scheme
(let* ((var1 10)
    (var2 (+ var1 12)))
;; But the definition of var1 could not refer to var2
)
```

نوع دیگر، “letrec”، طراحی شده تا رویه‌های بازگشتی دوگانه را قادر سازد به یکدیگر بسته شوند.
```scheme
;; Calculation of Hofstadter's male and female sequences as a list of pairs

(define (hofstadter-male-female n)
 (letrec ((female (lambda (n)
      (if (= n 0)
  1
  (- n (male (female (- n 1)))))))
    (male (lambda (n)
    (if (= n 0)
        0
        (- n (female (male (- n 1))))))))
 (let loop ((i 0))
 (if (> i n)
   '()
   (cons (cons (female i)
       (male i))
 (loop (+ i 1)))))))

(hofstadter-male-female 8)

===> ((1 . 0) (1 . 0) (2 . 1) (2 . 2) (3 . 2) (3 . 3) (4 . 4) (5 . 4) (5 . 5))
```

(دنباله‌های مرد و زن هوفستادر برای تعاریف مورد استفاده در این مثال را ببینید)

تمامی رویه‌های محدود در یک “letrec”، می‌توانند با نام به دیگری اشاره کنند، همچنین مقادیر متغیرهایی که قبلاً در همان “letrec” تعریف شدند، اما ممکن است به مقادیر تعریف شده بعدی در همان “letrec” اشاره نکنند.

یک نوع از “let”، فرم "named let"، یک شناسه پس از کلیدواژهٔ “let” دارد. این متغیرهای "let" را به آرگومان‌های یک رویه که نامش همان "شناسهٔ داده شده" و بدنه آن همان بدنه فرم "let" است، به هم می‌پیوندد (مقید می‌کند). بدنه می‌تواند با فراخوانی رویه، به دلخواه تکرار شود. "named let" به‌طور گسترده‌ای برای پیاده‌سازی تکرار استفاده می‌شود.

مثال: یک شمارنده ساده
```scheme
(let loop ((n 1))
 (if (> n 10)
 '()
 (cons n
     (loop (+ n 1)))))

===> (1 2 3 4 5 6 7 8 9 10)
```

مانند هر رویه در اسکیم، رویه ایجاد شده در “named let”، یک شی کلاس اول است.

## بازگشت دنباله مناسب

برای جزئیات بیشتر در این موضوع، بازگشت دنباله را ببینید.

اسکیم یک ساختار تکرار دارد، اما استفاده از بازگشت دنباله بیشتر اصطلاحی است. پیاده‌سازی استاندارد اسکیم برای بهینه‌سازی فراخوانی‌های دنباله به منظور پشتیبانی از تعداد نامحدودی از فراخوانی‌های فعال دنباله مورد نیاز هستند(R5RS بخش ۳٫۵). یک ویژگی که اسکیم به عنوان بازگشت دنباله مناسب توصیف می‌کند _ و آن را برای برنامه نویسان اسکیم برای نوشتن الگوریتم‌های تکراری با استفاده از ساختارهای بازگشتی آسان‌تر می‌کند، که بعضی اوقات بیشتر بصری و حسی هستند. رویه‌های بازگشت دنباله و فرم “named let”، از خاصیت «تکرار» با استفاده از بازگشت دنباله

پشتیبانی می‌کنند.
```scheme
;; Building a list of squares from 0 to 9:
;; Note: loop is simply an arbitrary symbol used as a label. Any symbol will do.

(define (list-of-squares n)
 (let loop ((i n) (res '()))
 (if (<i 0)
 res
 (loop (- i 1) (cons (* i i) res)))))

(list-of-squares 9)
===> (0 1 4 9 16 25 36 49 64 81)
```


## استانداردهای پیاده‌سازی
این زیربخش تصمیم‌های طراحی را مستند می‌کند که در طول سال‌ها به اسکیم یک شخصیت خاص داده‌اند، اما نه از نتایج اصلی طراحی.

### برج عددی
مقاله اصلی: Numerical tower

اسکیم یک مجموعه نسبتاً کامل از نوع داده‌های عددی را شامل انواع پیچیده و منطقی مشخص می‌کند که در اسکیم به عنوان برج عددی شناخته می‌شوند. (R5RS sec. 6.2).. این استاندارد، یا این‌ها مانند انتزاع برخورد می‌کند، و پیاده‌ساز را به هیچ نمایش خاص داخلی متعهد نمی‌کند. اعداد ممکن است کیفیت دقیق بودن را داشته باشند. یک عدد دقیق تنها می‌تواند توسط یک دنباله از عملیات دقیق که با دیگر اعداد دقیق درگیر هستند تولید شود—بنابریان نادقیق بودن واگیری است. استاندارد مشخص می‌کند که هر یک از دو پیاده‌سازی باید نتایج یکسانی را برای تمام عملیات بر روی اعداد دقیق داشته باشند.

استاندارد R5RS، روال exact->inexact و inexact->exact را مشخص می‌کند که می‌تواند در تغییر دقیق بودن یک عدد مورد استفاده قرار گیرد. رویه inexact->exact «عدد دقیقی که از لحاظ عددی نزدیکترین به آرگومان است» را تولید می‌کند. رویه exact->inexact «عدد نادقیق که از لحاظ عددی به آرگومان نزدیکترین است» را تولید می‌کند. استاندارد R6RS این رویه‌ها را از گزارش اصلی حذف می‌کند، اما آن‌ها را به عنوان رویه‌های سازگار R5RS در کتابخانه استاندارد مشخص می‌کند (rnrs r5rs (6)). در استاندارد R5RS، پیاده‌سازی‌های اسکیم نیاز به پیاده‌سازی کل برج عددی ندارند، بلکه آن‌ها باید «یک زیر مجموعه منسجم شامل هردوی مقصود پیاده‌سازی و روح زبان اسکیم» را پیاده‌سازی کنند. (R5RS sec. 6.2.3).. استاندارد R6RS جدید نیاز به پیاده‌سازی تمام برج دارد، و «شی‌های عدد صحیح دقیق و شی‌های عدد منطقی دقیق از عملاً اندازه نامحدود و دقت، و پیاده‌سازی کردن رویه‌های مخصوص… که آن‌ها همیشه نتایج دقیقی را هنگامی که آرگومان‌های دقیق به آن‌ها می‌دهیم بازمی‌گردانند» (R6RS sec. 3.4, sec. 11.7.1).

مثال ۱: ریاضیات دقیق در یک پیاده‌سازی که اعداد پیچیده منطقی دقیق را پشتیبانی می‌کند.
```scheme
;; Sum of three rational real numbers and two rational complex numbers

(define x (+ 1/3 1/4 -1/5 -1/3i 405/50+2/3i))

x

===> 509/60+1/3i

;; Check for exactness.

(exact? x)

===> #t
```


مثال ۲: همان ریاضی در یک پیاده‌سازی که هیچ‌یک از اعداد منطقی یا پیچیده را پشتیبانی نمی‌کند ما اعداد حقیقی را در نشانه منطقی می‌پذیرد: 
```scheme
;; Sum of four rational real numbers

(define xr (+ 1/3 1/4 -1/5 405/50))

;; Sum of two rational real numbers

(define xi (+ -1/3 2/3))

xr

===> 8.48333333333333

xi

===> 0.333333333333333

;; Check for exactness.

(exact? xr)

===> #f

(exact? xi)

===> #f
```


هر دو پیاده‌سازی با استاندارد R5RS تأیید می‌شوند اما دومی با استاندارد R6RS تأیید نمی‌شود زیرا تمام برج اعداد را پیاده‌سازی نکرده‌است.

### ارزیابی تأخیر
همچنین ببینید: Lazy evaluation

اسکیم ارزیابی تأخیر را پشتیبانی می‌کند در قالب delay و رویه force.
```scheme
(define a 10)

(define eval-aplus2 (delay (+ a 2)))

(set! a 20)

(force eval-aplus2)

===> 22

(define eval-aplus50 (delay (+ a 50)))

(let ((a 8))

 (force eval-aplus50))

===> 70

(set! a 100)

(force eval-aplus2)

===> 22
```


متن نحوی تعریف اصلی از قول حفظ شده‌است، و ارزش آن نیز پس از استفاده از force حفظ شده‌است. قول تنها یکبار ارزیابی می‌شود. این نخستین‌ها، که مقادیر شناخته را یه عنوان قول تولید یا رسیدگی می‌کنند، می‌توانند برای ساختارهای پیاده‌سازی ارزیابی تنبل پیشرفته مانند جریان‌ها [۲۰] استفاده شوند. در استاندارد R6RS، این‌ها دیگر اولیه نیستند، اما در عوض، به عنوان بخشی از کتابخانه سازگاری R5RS (rnrs r5rs (6)) فراهم می‌شوند. در R5RS، یک پیاده‌سازی پیشنهاد شده از delay و force، پیاده‌سازی قول به عنوان یک رویه بدون آرگومان (یک تانک) و استفاده از یادداشت برداری برای اطمینان از این که تنها یکبار ارزیابی صورت گرفته‌است، صرف نظر از تعداد دفعاتی که force صدا زده می‌شودد (R5RS sec. 6.4).[3] داده می‌شود. SRFI 41 اصطلاح هر دوی دنباله‌های محدود و نامحدود را با صرفه جویی خارق‌العاده‌ای توانمند می‌کند. برای مثال، این یک تعریف از دنباله فیبوناچی با ساتفاده از توابع تعریف شده در SRFI 41:[20] است.
```scheme
;; Define the Fibonacci sequence:

(define fibs

 (stream-cons 0

 (stream-cons 1

 (stream-map +

 fibs

 (stream-cdr fibs)))))

;; Compute the hundredth number in the sequence:

(stream-ref fibs 99)

===> 218922995834555169026
```


### ترتیب ارزیابی آرگومان‌های رویه
بیشتر Lispها یک ترتیب را بای ارزیابی آرگومان‌های رویه مشخصی می‌کنند. ترتیب ارزیابی‌ها—شامل ترتیب در اینکه عبارت‌ها در موقعیت عملیات ارزیابی شده‌اند—ممکن است توسط یک پیاده‌سازی روی یک پایه فراخوانی-بوسیله-فراخوانی انتخاب شده باشد، و تنها محدودیت این است که "تاثیر هرکدام از ارزیابی‌های موازی عبارات عملگر و عمولند، محدود است تا با برخی دنباله ترتیب ارزیابی استوار شود." (R5RS sec. 4.1.3)[3]
```scheme
(let ((ev (lambda(n) (display "Evaluating ")

 (display (if (procedure? n) "procedure" n))

 (newline) n)))

 ((ev +) (ev 1) (ev 2)))

===> 3
```


Ev یک رویه است که آرگومان‌های داده شده به آن را بررسی می‌کند، سپس مقدار آن را برمی‌گرداند. در تضاد با Lispهای دیگر، ظاهر یک عبارت در در موقعیت عملگر (اولین گزینه) از یک عبارت اسکیم، تا زمانی که نتیجه عبارت یک رویه در یک موقعیت عملکر دیگر است، کاملاً قانونی است.

در فراخوانی رویه "+" برای جمع ۱ و ۲، عبارت (ev +), (ev 1) و (ev 2) ممکن است تا زمانی که ارزیابی موازی آن‌ها بر روی هم تأثیری نداشته باشند، بدون هیچ ترتیبی ارزیابی شوند؛ بنابراین این سه خط که در ادامه آورده شده‌است، ممکن است در هیچ ترتیبی توسط استاندارد اسکیم زمانی ه کد مثال بالا اجرا می‌شود نمایش داده نشوند، اگرچه متن یک خط ممکن است با دیگر عوض نشده باشد بخاطر اینکه ممکن است دنباله تمرکز ارزیابی را نقص کند.

ارزیابی ۱

ارزیابی ۲

ارزیابی رویه

### ماکروهای بهداشتی
مقاله اصلی: Hygienic macro

در استاندارد R5RS و همچنین گزارش‌های بعد از آن، نحو اسکیم می‌تواند به آسانی توسط سیستم‌های ماکرو مشتق شود. استاندارد R5RS یک ماکرو بهداشتی قدرتمند را معرفی می‌کند که به برنامه نویسان اجازه می‌دهد تا ساختار نحوی جدید را به زبان با استفاده از الگوهای ساده که با زیرزبان همخوانی دارد اضافه کنند. (R5RS sec 4.3).[3]. با توجه به این، سیستم ماکرو بهداشتی به یک ضمیمه از استاندارد R4RS، یه عنوان یک سیستم سطح بالا همراه با سطح با سیستم ماکرو سطح پایین کاهش یافته‌است، که با هر دوی آن‌ها مانند افزونه‌هایی به اسکیم رفتار می‌شود تا یک بخش ضروری از زبان [۲۱].

پیاده‌سازی سیستم بهداشتی ماکرو، که همچنین syntax-rules نیز می‌گویند، باید دامنه نحوی باقی‌مانده زبان را رعایت کنند؛ که این کار با یک نام‌گذاری خاص و قواعد دامنه برای گسترش ماکرو و اجتناب از خطاهای برنامه‌نویسی رایج که می‌توانند در سیستم ماکرو دیگر زبان‌های برنامه‌نویسی رخ بدهند، خاطر جمع می‌شود. R6RS سیستم‌های تحول پیچیده بیشتری را مشخص می‌کند، syntax-case، که به عنوان یک افزونه برای اسکیم R5RS برای مدتی موجود بوده‌است.
```scheme
;; Define a macro to implement a variant of "if" with a multi-expression

;; true branch and no false branch.

(define-syntax when

 (syntax-rules ()

 ((when pred exp exps ...)

 (if pred (begin exp exps ...)))))
```


دعوت‌ها از ماکروها و رویه‌ها یک شباهت نزدیک دارند – هر دو s-عبارت هستند—اما با آن‌ها متفاوت رفتار می‌شود. وقتی کامپایلر به یک s-عبارت در برنامه برخورد می‌کند، اول چک می‌کند تا بفهمد که آیا علامت به عنوان یک کلید نحوی درون دامنه نحوی حال حاضر قرار دارد یا خیر. اگر قرار دارد، سپس تلاش می‌کند تا ماکرو را گسترش دهد، با گزینه‌های دنباله s-عبارت مانند آرگومان‌ها بدون کد کامپیایل برای ارزیابی کردن رفتار می‌کند، و این فرایند به صورت بازگشتی تکرار می‌شود تا هیچ ماکرو دعوت دیگری باقی نماند. اگر یک کلید نحوی نباشد، کامپایلر کد را برای ارزیابی آرگومان درون دنباله s-عبارت و سپس ارزیابی متغیر معرفی شده توسط علامت در هد s-عبارت کامپایل می‌کند و آن را به عنوان یک رویه با دنباله ارزیابی عبارات با آرگومان‌های واقعی برای آن‌ها فراخوانی می‌کند.

بیشتر پیاده‌سازی‌های اسکیم همچنین سیستم‌های ماکرو اضافی ارائه می‌دهند. میان آنها، محبوب ترین‌ها syntactic closures, explicit renaming macros و define-macro یک ماکرو سیستم غیر بهداشتی شبیه به سیستم فراهم شده در Lisp رایج هستند.

### محیط‌ها و ارزیابی
با توجه به R5RS، اسکیم هیچ استانداردی برای رویه ارزیابی که در همه Lispهای دیگر وجود دارد، ندارد، اگرچه اولین مقاله لامبدا ارزیابی را "شبیه به تابع ارزیابی Lisp" توصیف کرده بود. [۱۶] و اولین گزارش تجدید نظر شده در ۱۹۷۸ آن را با enclose عوض کرد، که دو آرگومان را می‌گیرد. دومی، سومی و چهارمین گزارش تجدید نظر شده، هر گونه معادل بودن ارزیابی را حذف کردند.

دلیل این گیجی این است که در اسکیم با دامنه نحوی نتایج ارزیابی یک عبارت بستگی به جایی دارد که ارزیابی می‌شود. برای مثال، واضح نیست که نتیجه از ارزیابی عبارت زیر باید ۵ باشد یا 6: [22]
```scheme
(let ((name '+))

 (let ((+ *))

 (evaluate (list name 2 3))))
```


اگر در محیط بیرونی تر ارزیابی شود، جایی که name تعریف شده‌است، نتیجه جمع عملگر هاست. اگر در محیط درونی ارزیابی شود، جایی که علامت "+" است، نتیجه ضرب دو عملگر خواهد بود.

R5RS این گیجی را با استفاده از مشخص کردن سه رویه که محیطها را باز میگردانندو فراهم کردن یک رویه ارزیابی که یک s-عبارت و یک محیط را می‌گیرد و عبارت را درمحیط فراهم شده ارزیابی می‌کند حل می‌کند. (R5RS sec. 6.5)[3]. R6RS این را به وسیلهٔ فراهم کردن یک رویه به نام environment که یک برنامه‌نویس می‌تواند دقیقاً مشخص کند کدام شی‌ها را وارد به محیط ارزیابی وارد کند مشتق می‌کند.

### رفتار مقادیر غیر بول در عبارات بول
در بسیاری از لهجه‌های Lisp، شامل Lisp رایج، با قرار دادن مقدار NIL در یک مقدار در یک عبارت بول، آن را غلط تعیین می‌کنند. در اسکیم از زمان استاندارد IEEE در ۱۹۹۱، [۲] تمام مقادیر بجز #f، شامل NILهای معادل در اسکیم که مانند '() نوشته می‌شوند، به عنوان مقدار درست در عبارات بول ارزیابی می‌شوند. (R5RS sec. 6.3.1)[3]

از آنجا که در بیشتر Lispها مقدار صحیح T می‌باشد، ثابتی که مقدار درست را در اسکیم نمایندگی می‌کند با #T نشان می‌دهیم.

### عدم هماهنگی نوع داده‌های اولیه

### معادله اساسی
همچنین بینید: relational operator

سه نوع تعادل میان شی‌های خودسرانه در اسکیم وجود دارد که با سه روش معادله اساسی مختلف، عملگرهای ارتباطی برای امتحان کردن تساوی eq?, eqv? و equal? مشخص شده‌اند:

- eq? به #f ارزیابی می‌شود مگر اینکه پارامترهایش نماینده داده یکسان در حافظه باشند؛ eqv? در کل شبیه به همان eq? است اما با داد ه‌های اولیه کار می‌کند (مانند کاراکترها و اعداد) خصوصااعدادی که همان مقداری که را نمایندگی می‌کنند که eqv? هستند حتی اگر به همان شی اشاره نکنند.

- equal? دو ساختار داده را مانند لیست‌ها، بردارها و رشته‌ها مقایسه می‌کند تا مشخص شود آن‌ها ساختار متضاد و محتوای eqv? دارند. (R5RS sec. 6.1)[3]

وابستگی عملیات داده همچنین در اسکیم نیز وجود دارد: string=? و string-ci=? دو رشته را مقایسه می‌کنند (دومی یک مقایسه وابسته به بزرگی و کوچکی حروف را انجام می‌دهد); char=? و char-ci=?دو کاراکتر را مقایسه می‌کنند؛ = دو عدد را مقایسه می‌کند. [۳]

### پیام‌ها
همچنین ببینید: Comment (computer programming)

براساس استاندارد R5RS، پیام استاندارد در اسکیم یک نقطه ویرگول بود، که بقیه خط‌ها را برای اسکیم نامریی می‌کرد. بسیار از پیاده‌سازی‌ها کنوانسیون‌های اجازه دهنده به پیام‌گذاری را پشتیبانی می‌کنند که برای بیش از یک خط باشد، و استاندارد R6RS هر دوی آن‌ها را اجازه می‌دهد: یک s-عبارت می‌تواند تماماً با یک علامت #; به یک پیام تبدیل شود و یک دامنه پیام می‌تواند با احاطه شدن توسط علامت‌های #| و |# تولید شود.

### ورودی / خروجی
ورودی و خروجی اسکیم بر درگاه نوع داده بنا شده‌است. (R5RS sec 6.6)[3]. R5RS دو درگاه پیش‌فرض قابل دستیابی بت رویه‌های current-input-port و current-output-port را تعریف می‌کند، که با مفاهیم ورودی استاندارد و خروجی استاندارد در یونیکس مطابق دارند. بیشتر پیاده‌سازی‌ها current-error-port را نیز فراهم می‌آورند. تغییر مسیر ورودی و خروجی استاندارد در استاندارد، به وسیلهٔ رویه‌های استاندارد مانند with-input-from-file و with-output-to-file پشتیبانی می‌شود. بیشتر پیاد هسازی‌ها درگاه‌های رشته را با قابلیت‌های شبیه به تغییر مسیر، توامند کردن بسیاری از عملیات ورودی-خروجی که به جای فایل بافر باشند را به وسیلهٔ رویه توصیف شده در SRFI 6.[24] فراهم می‌آورند. استاندارد R6RS رویه‌های درگاه پیچیده‌تر و تواناتر و بسیاری دیگر از انواع جدید درگاه را فراهم می‌آورند. مثال‌های زیر در R5RS سختگیرانه اسکیم نوشته شده‌اند.

مثال ۱: با خروجی پیشفرض به جریان خروجی درگاه
```scheme
(let ((hello0 (lambda() (display "Hello world") (newline))))

 (hello0))
```


مثال ۲: مانند ۱، اما با استفاده از آرگومان درگاه اختیاری به رویه خروجی
```scheme
(let ((hello1 (lambda (p) (display "Hello world" p) (newline p))))

 (hello1 (current-output-port)))
```


مثال ۳: مانند ۱، اما خروجی به یک فایلی که جدید ساخته شده‌است منتقل شده
```scheme
;; NB: with-output-to-file is an optional procedure in R5RS

(let ((hello0 (lambda () (display "Hello world") (newline))))

 (with-output-to-file "helloworldoutputfile" hello0))
```


مثال ۴: مانند ۲، اما با بازکردن و بستن درگاه صریح فایل به ارسال خروجی به فایل
```scheme
(let ((hello1 (lambda (p) (display "Hello world" p) (newline p)))

 (output-port (open-output-file "helloworldoutputfile")))

 (hello1 output-port)

 (close-output-port output-port))
```


مثال ۵: مانند ۲، اما با ستفاده از فراخوانی-با-فایل-خروجی به ارسال خروجی به فایل
```scheme
(let ((hello1 (lambda (p) (display "Hello world" p) (newline p))))

 (call-with-output-file "helloworldoutputfile" hello1))
```


رویه‌های مشابه برای ورودی فراهم شده‌اند. R5RS مسندهای input-port? و output-port? فراهم می‌کند. برای ورودی و خروجی کاراکتر write-char, read-char, peek-char و char-ready? فراهم شده‌اند. برای خواندن و نشوتن عبارات اسکیم، اسکیم read و write را فراهم کرده‌است. در یک عملیات خواندن، نتیجه باگردانده شده پایان-شی-فایل است اگر درگاه ورودی به آخر فایل رسیده باشد، و این می‌تواند توسط مسند eof-object? آزماش شود.

علاوه بر استاندارد، SRFI 28 یک فرمت پایه تعریف شده شبیه به تابع فرمت لیسپ رایج، بعد از اینکه نامگذاری شد را تعریف می‌کند. [۲۵]

### باز تعریف رویه استاندارد
در اسکیم رویه‌ها محدود به متغیرها هستند. در استاندارد R5RS زبان کاملاً مجاز است که اتصالات ساخته شده طی رویه متغیر را تغییر دهد، و به‌طور مؤثر آن‌ها را بازتعریف کند. (R5RS "تغییرهای زبان")[۳]. برای مثال، ممکن است + را طوری گسترش دهد که رشته را قبول کند همان‌طور که برای اعداد عمل می‌کند:
```scheme
(set! +

 (let ((original+ +))

 (lambda args

 (if (and (not (null? args)) (string? (car args)))

 (apply string-append args)

 (apply original+ args)))))

(+ 1 2 3)

===> 6

(+ "1" "2" "3")

===> "123"
```


در R6RS هر اتصال، شامل اتصالات استاندار، به یک کتابخانه تعلق دارند و تمامی اتصالات خروجی تغییرناپذیر هستند. (R6RS sec 7.1)[4]. به همین خاطر، بازتعریف رویه‌های استاندارد به وسیله جهش ممنوع است. در عوض، این امکان وجود دارد که رویه‌های مختلف تحت نام‌های رویه‌های استاندارد وارد شوند، که اثر آن شبیه به بازتعریف است.

بررسی رویه‌ها و فرم‌های استاندارد

زبان رسمی در استانداردهای R5RS (1998) و R6RS (2007) تعریف شده‌است. آن‌ها یک «فرم» استاندارد توصیف می‌کنند: کلمات کلیدی و نحو همراه، که ساختار کنترل زبان را ارائه می‌دهند و رویه‌های استاندارد که کارهای معمول را انجام می‌دهند.

## فرمهای استاندارد
این جدول فرم‌ها و اَشکال استاندارد در اسکیم را توصیف می‌کند. بعضی از فرم‌ها در بیش از یک ردیف ظاهر می‌شوند زیرا آن‌ها به راحتی نمی‌توانند به یک تابع در زبان طبقه‌بندی شوند. فرم‌ها با نماد "L" در این جدول به عنوان فرم‌های "کتابخانه" طبقه‌بندی شده و اغلب با استفاده از فرم‌های بنیادی تر به عنوان ماکروها پیاده‌سازی می‌شوند؛ که باعث می‌شود وظیفه پیاده‌سازی بسیار ساده‌تر از سایر زبان‌ها شود.

فرم‌های استاندارد در زبان اسکیم R5RS

توجه داشته باشید که “begin” به عنوان یک نحو کتابخانه در R5RS تعریف می‌شود، اما توسعه دهنده باید برای دستیابی به قابلیت پیوند، در مورد آن بداند. در R6RS، دیگر یک نحو کتابخانه نیست.

## رویه‌های استاندارد

دو جدول زیر رویه‌های استاندارد در اسکیم R5RS را توصیف می‌کند. R6RS بسیار گسترده‌تر است و خلاصه‌ای از این نوع هم کارا نخواهد بود.

بعضی رویه‌ها در بیشتر از یک خط ظاهر می‌شوند زیرا به راحتی نمی‌توانند به یک تابع در زبان طبقه‌بندی شوند.

رویه‌های استاندارد در زبان اسکیم R5RS

رویه‌های کاراکتری و رشته‌ای که "-ci" را در نام خود دارند، مقایسه‌های مستقل موردی بین آرگومانهایشان انجام می‌دهند: نسخهٔ حروف بزرگ و حروف کوچک از یک کاراکتر یکسان، برابر خواهد بود.

رویه‌های عددی استاندارد در زبان اسکیم R5RS

پیاده‌سازی – و / که بیش از دو آرگومان را تعریف می‌کند اما در R5RS اختیاری است.

## درخواست‌های طرح برای اجرا (پیاده‌سازی)
به دلیل کوچک‌سازی طرح، بسیاری از رایج‌ترین روش‌ها و صورت‌های نحوی به وسیله استاندارد تعریف نمی‌شود. به منظور کوچک نگهداشتن زبان هسته‌ای (اصلی) اما استانداردسازی پسوندها، جامعه طرح دارای درخواست طرح برای فرایند اجراست (srfi) که بدان وسیله مجموعه‌های پسوند از طریق بحث دقیق طرح‌های پیشنهادی پسوند تعریف می‌شوند. این امر قابلیت انتقال کد را افزایش می‌دهد. بسیاری از srfiها به وسیلهٔ همه اجراها تا بیشتر اجراهای طرح پشتیبانی می‌شوند. برخی srfiهایی که دارای پشتیبانی نسبتاً گسترده در اجراهای متفاوت هستند، شامل موارد زیر می‌باشند:

-ساخت پسوند شرطی مبتنی بر ویژگی

-مجموعه فهرست

-انواع داده‌های برداری عددی همگن

-پورت رشته اولیه

-دریافت و پیوند با مقادیر متعد

-تعریف انواع ثبت

-مجموعه رشته

## پیاده‌سازی‌ها
دسته اصلی: Scheme (programming language)

طراحی کمینه و ظریف اسکیم را به یک هدف محبوب برای طراحان زبان، دنبال کنندگان سرگرمی و آموزنده‌ها تبدیل کرده‌است و بخاطر اندازه کوچکش به دلیل مفسر کوچک، همچنین یک انتخاب محبوب برای سیستم‌های جاسازی شده و اسکریپتی است. این نتیجه از امتیازات پیاده‌سازی بدست آمده‌است. که بسیاری از آن‌ها با هم متفاوت هستند بنابرین حمل برنامه‌ها از یک پیاده‌سازی به پیاده‌سازی‌های دیگر کاملاً دشوار است و اندازه وچک زبان استاندارد به این معناست که نوشتن یک برنامه مفید با هر پیچیدگی دراسکیم استاندارد و قابل حمل تقریباً غیرممکن است. استاندارد R6RS یک زبان خیلی گسترده‌تر را طی تلاشی برای گسترده‌تر کردن درخواست‌های برنامه نویسانش مشخص می‌کند. تقریباً تمامی پیاده‌سازی یک حالت-لیسپ سنتی حلقه خواندن-ارزیابی-چاپ را برای توسعه و خطایابی ارائه می‌دهند. همچنین اغلب، برنامه‌های اسکیم را به اتصالات قابل اجرا تبدیل می‌کنند. پشتیبانی برای کد اسکیم تعبیه شده در برنامه‌های نوشته شده به دیگر زبان‌ها نیز مانند سادگی نسبی پیاده‌سازی‌های اسکیم که آن را یک انتخاب محبوب برای اضافه کردنقابلیت‌های اسکریپتی به سیستم‌های بزرگتر توسعه یافته در زبان‌هایی مانند c کرده‌است، رایج است.

Gambit, Chicken, و Biglooبا اسکیمای کامپایل شونده به c کار می‌کنند که به‌طور خاص تعبیه کردن را آسان می‌کنند. علاوه بر این، مترجم Bigloo می‌تواند جوری پیکربندی شود که بایت کدهای JVM تولید کند، و همچنین یک تولیدکننده بایت کد برای .NET باشد.

برخی پیاده‌سازی‌ها ویژگی‌های اضافه تری را فراهم می‌کنند. برای مثال Kawa و JSchee یکپارچه‌سازی با کلاس‌های جاوا را فراهم می‌کنند، و اسکیم به مترجم‌های c معمولاً کار راب برای کتابخانه‌های خارجی نوشته شده به زبان c، بسته به اینکه به کد تعبیه شده در c واقعی در سورس اسکیم اجازه دهیم، راحت می‌کند. یک مثال دیگر Pvts است، که یک مجموعه از ابزار بصری را برای پشتیبانی یادگیری اسکیم ارائه می‌دهد.

## کاربردها
این طرح به‌طور گسترده به وسیلهٔ چندین مدرسه مورد استفاده قرار می‌گیرد. به‌طور ویژه دوره‌های علوم کامپیوتری مقدماتی از این طرح در رابطه با ساختار کتاب درسی و تغییر برنامه‌های کامپیوتری (sicp) استفاده می‌کنند. برای ۱۲ دهه گذشته plt پردازی pragrambydesign اجرا کرده‌است.

2- mit ۶٬۰۰۱ قدیمی در این طرح آموزش داده شده. هر چند ۶٬۰۰۱ با دوره‌های جدید تر جایگزین شد، اما هنوز sicp آموزش به زبان mit را ادامه می‌دهد. کتاب دو چگونه برنامه‌ها را طراحی کنیم (با نویسندگی مایتاس فلیس) به وسیلهٔ چندین مؤسسه آموزش عالی برای دوره‌های علوم کامپیوتری مقدماتی آن‌ها به کار برده می‌شود. هم دانشگاه فورس استرن و هم مؤسسه ورسسترپلی تکنیک از این طرح به‌طور انحصاری برای دوره‌های مقدماتی «اصول علوم کامپیوتری» و مقدمه‌ای برطراحی برنامه استفاده می‌کنند رز_هولمان از این طرح در دوره مفاهیم زبان برنامه‌نویسی پیشرفته تر استفاده می‌کند.

## ویژگی‌ها
روالها در اسکیم، تابع‌های دستهٔ اول هستند که برنامه‌نویسی تابعی را شدنی می‌کند.

## منابع

* مشارکت‌کنندگان ویکی‌پدیا. «Scheme (Programming language)». در دانشنامهٔ ویکی‌پدیای انگلیسی.



رده:اسکیم (زبان برنامه‌نویسی)
رده:زبان‌های برنامه‌نویسی آکادمیک
رده:زبان‌های برنامه‌نویسی ساخته‌شده در ۱۹۷۵ (میلادی)
رده:زبان‌های برنامه‌نویسی ساخته‌شده در دهه ۱۹۷۰ (میلادی)
رده:زبان‌های برنامه‌نویسی ماشینی پویا
اسکیم
رده:مقاله‌های دارای کد مثال اسکیم (زبان برنامه‌نویسی)
رده:زبان‌های برنامه‌نویسی با نحو گسترش‌پذیر